

A timed IO monad

David Janin
Bordeaux INP, UMR CNRS LaBRI,
University of Bordeaux
May 2019

The need for a timed monad

Music playing example

Plays every second n the note $s\ n$ for half a second.


What is wrong with that code ?

```
perform :: Int → IO ()  
perform n = do { play (s n) (5 * 105); perform (n + 1) }  
where  
  play n d = noteOn n >> threadDelay d >>  
             noteOff n >> threadDelay d
```

The need for a timed monad

Every computation takes some time,
every delay resumes too late

Specified vs actual scheduling



Time leak

Caused by a positive (good) but unbounded (bad) time drift defined, at any instant, by:

$$\textit{time drift} = \textit{actual timestamp} - \textit{specified timestamp}$$

The need for a timed monad

General specifications

A timed monad shall provide:

- (1) **specified** positive or zero duration for each action,
- (2) induced **expected scheduling** of actions,
- (3) performed **actual scheduling** of actions,
- (4) tools to control/observe (positive) **time drift**, that is, the (miss)match between expected and actual scheduling,
- (5) an other useful features such as safe and robust **IO**, **concurrency**, ...

Side remarks

- (1) passing time *is* a side effect,
- (2) every programmer *knows* monad programming style.

The need for a timed monad

Back to our exemple

Within the default timed IO monad, we would **essentially** write the same code:

```
perform :: Int → TIO ()
perform n = do { play (s n) (5 * 10^5); perform (n + 1) }
  where
    play n d = liftIO (noteOn n)           -- specified duration 0
              >> delay d                  -- specified duration d
              >> liftIO (noteOff n)      -- specified duration 0
              >> delay d                  -- specified duration d
```

but, in order to compensate the time drifts created by all other computations, the **actual duration** of *delay d* **shorter than** its **specified duration** of *d* microseconds.

Structure of the talk

Timed monad

Monad with references

Monad streams

Monad stream references

Conclusion

1. Timed monad

Timed programming with class

Timed Monad type class

```
newtype Time d = Time d deriving (Eq, Ord)
shift (Time t) d = Time (t + d)
duration (Time t1) (Time t2) = t1 - t2
```

with **timestamp type** ($\text{Time } d$) and **duration type** (d),

```
class (Monad m, MonadIO m, Num d, Ord d)
  => TimedMonad m d | m → d where
  runTIO :: m a → IO a --
  delay :: d → m () -- negative d treated as zero
  now :: m (Time d) -- current specified timestamp
  drift :: m d -- current actual time drift
```

with all timed actions specified as **instantaneous** but *delay* d specified with **duration** d , ..., and with *delay* d that resumes as soon as the new **specified** timestamp is **actually** passed **for real**.

Timed Monad type class

The **specified duration** of an action is computed by:

```
dur m = do { t0 ← now; _ ← m;
            t1 ← now; return (duration t1 t0) }
```

In addition to superclasses laws, we shall have:

$$\mathit{dur} \, m \equiv \mathit{dur} \, (\mathit{fmap} \, f \, m) \quad (1)$$

$$\mathit{dur} \, (\mathit{return} \, a) \equiv \mathit{return} \, 0 \quad (2)$$

$$\mathit{dur} \, (m_1 \gg m_2) \equiv \mathit{dur} \, (m_1) \gg \lambda d \rightarrow \mathit{fmap} \, (d+) \, (\mathit{dur} \, m_2) \quad (3)$$

for every timed action m , m_1 and m_2 , and

$$\mathit{dur} \, (\mathit{liftIO} \, io) \equiv \mathit{liftIO} \, io \gg \mathit{return} \, 0 \quad (4)$$

$$io \equiv (\mathit{runTIO} \circ \mathit{liftTIO}) \, io \quad (5)$$

for every IO action io , therefore $m \neq (\mathit{liftIO} \circ \mathit{runTIO}) \, m$ whenever the timed action m has **non zero duration**.

Timed lifting of IO actions

As an example of a derived function:

A timed lifting of an IO action:

```
liftTimedIO :: TimedMonad m d  $\Rightarrow$  IO a  $\rightarrow$  m a  
liftTimedIO m = do  
  a  $\leftarrow$  liftIO m      -- specified duration 0  
  drift  $\gg=$  delay      -- specified duration : the actual time drift  
  return a              -- here time drift is nearly zero
```

where, upon exit, the expected timestamp essentially matches the specified time stamp, since the **actual duration** of *drift $\gg=$ delay* shall be nearly zero.

A timed IO monad instance

Define **timed IO monad states** as (implicit) IO monad states extended with (explicit) timestamp, therefore, **timed IO actions** by:

```
newtype TIO a = TIO (Time Integer → IO (Time Integer, a))
```

with expected functor, monad and monadIO instances:

```
instance Functor TIO where
```

```
  fmap f (TIO g) = TIO $ λs → do  
    (t, a) ← g s  
    return (t, f a)
```

```
instance Monad TIO where
```

```
  return a = TIO $ λs → return (s, a)  
  (≫=) (TIO f) g = TIO $ λs → do  
    (t, a) ← f s  
    let TIO h = g a in h t
```

```
instance MonadIO TIO where
```

```
  liftIO m = TIO $ λs → m ≫= λa → return (s, a)
```

The timed IO monad instance

Then we put:

```
instance TimedMonad TIO Integer where
  now = TIO ( $\lambda s \rightarrow$  return (s, s))
  drift = TIO ( $\lambda s \rightarrow$  systemTime  $\gg=$ 
                $\lambda r \rightarrow$  return (s, duration r s))
  delay d = TIO $  $\lambda s \rightarrow$  if (d  $\leq$  0) then (return (s, ()))
             else $ do
    let t = shift s d           -- expected timestamp target
        r  $\leftarrow$  systemTime   -- current actual timestamp
        threadDelay (duration t r) -- t just passed for real
    return (t, ())
  runTIO (TIO f) = systemTime  $\gg=$  f  $\gg=$  return  $\circ$  snd
```

with *systemTime* that returns system timestamp in **micro seconds**.



What we actually need for generic timed IO monads

The class type: $Timer\ s\ d \mid s \rightarrow d$

that defines and binds together:

- ▶ a **duration** type d , with derived timestamps and time scale,
- ▶ a possible external **scheduler**,
- ▶ a **timed state** type s with embedded expected timestamp and scheduler handle,
- ▶ the associated **runtime calls** $systemTime$, $systemDelay$ with (current) timed state argument,
- ▶ an **initialization action** $initialState$.

Timed IO monad instances (simple performance tests)



2. Monad with references

Safe communicating processes with class,


musng around ideas of Simon Marlow in [Control.Concurrent.Async](#).

On monad programming

When programming within a monad, there are:


- (1) **action programs** that can be freely reused, duplicated, in a pure functional programming style,
- (2) **action running instances** that cannot be reused, duplicated, etc. . . , for they are uniquely located in the underlying monad state space.

In every **strict monad**, function $bind :: m\ a \rightarrow (a \rightarrow m\ b) \rightarrow m\ b$ (implicitly) encodes two “functions” *actionExec* and *continueWith* performed in this order:



Monad references

A **monad reference** shall be a broadcast channel from a **uniquely** associated running monad action, which can be freely read, in a **safe** and **robust** way.



with, generalizing *async* library, the following class type:

```
class Monad m  $\Rightarrow$  MonadRef m where  
  type Ref m :: *  $\rightarrow$  *  
  forkToRef :: m a  $\rightarrow$  m (Ref m a)  
  readRef :: Ref m a  $\rightarrow$  m a  
  tryReadRef :: Ref m a  $\rightarrow$  m (Maybe a)  
  parReadRef :: Ref m a  $\rightarrow$  Ref m b  $\rightarrow$  m (Either a b)
```

Safely ? Robustly ?

Safely

Deadlock free communication schema for a process can only access the data produced by another if it “sees” its fork.

Robustly

No or harmless side effects when reading monad references for reading a process reference does not imply re-executing that process and its side effects.

Example in music (with actions replaced by streams of actions)

Given a playing music running process, the reference to that process could be the score of the played music. Reading the score does not imply replaying the music.

Monad reference laws

Monad with references instance shall satisfy the following laws:

- ▶ Basic semantics

$$m \equiv \text{forkToRef } m \gg\gg \text{readRef} \quad (6)$$

- ▶ Idempotence

$$\text{readRef } r \equiv \text{readRef } r \gg\gg \text{readRef } r \quad (7)$$

- ▶ Commutation

$$\begin{aligned} \text{readRef } r_1 \gg\gg \lambda x_1 \rightarrow \text{readRef } r_2 \\ \gg\gg \lambda x_2 \rightarrow \text{return } (x_1, x_2) \\ \equiv \text{readRef } r_2 \gg\gg \lambda x_2 \rightarrow \text{readRef } r_1 \\ \gg\gg \lambda x_1 \rightarrow \text{return } (x_1, x_2) \end{aligned} \quad (8)$$

Parallel reading

As a derived function:

```
parRun :: MonadRef m => m a -> m b -> m (Either a b)
parRun m1 m2 = do { r1 ← forkToRef m1; r2 ← forkToRef m2;
                    parReadRef r1 r2 }
```

with an induced race as illustrated by:

```
parRun (return "foo") (return "foo")
```

that returns either *Left* "foo" or *Right* "foo".

Timed IO references

Reference to timed IO actions are defined by:

```
newtype TIORef a =  
  TIORef (Time Integer, MVar (Timed Integer, a))
```

where, in *TIORef* (*s*, *v*), there shall be the **start timestamp** *s* with the mutable variable *v* filled with **stop timestamp** and **returned value** upon termination of the referenced running timed action.

```
instance MonadRef TIO where  
  type Ref TIO = TIORef  
  readRef (TIORef (_, v)) = TIO $ \s → do  
    (s1, a) ← readMVar v  
    return (max s s1, a)  
  forkToRef (TIO m) = TIO $ \s → do  
    v ← newEmptyMVar  
    _ ← forkIO (m s >>= putMVar v)  
    return (s, TIORef (s, v))  
  ....
```

with some more lines of code for *parReadRef*.

Timed monad with references

Combining timed monad with monad references yields:

```
class (TimedMonad m d, MonadRef m)  $\Rightarrow$   
  TimedMonadRef m d where  
    durRef :: Ref m a  $\rightarrow$  m d
```

with:

```
instance TimedMonadRef TIO Integer where  
  durRef (TIORef (s0, v)) = TIO $  $\lambda s \rightarrow$  do  
    (s1, _)  $\leftarrow$  readMVar v  
    return (max s s1,  
            duration (getStateTime s1) (getStateTime s0))
```

Replaying a timed monad reference

As application example, one can **replay** a referenced timed monad action with same returned value and **same duration** by:

```
delayRef :: TimedMonadRef m d ⇒ Ref m a → m ()  
delayRef r = do  
  t0 ← now  
  d ← durRef r  
  t1 ← now  
  delay (d - duration t1 t0)  
  
replayRef :: TimedMonadRef m d ⇒ Ref m a → m a  
replayRef r = delayRef r ≫ readRef r
```

Reading, replaying or expanding a forked process reference

- Forked process: $r \leftarrow \text{forkToRef} (\text{delay } 5)$



- Read reference: $\text{delay } 2 \gg \text{readRef } r$



- Replay reference: $\text{delay } 2 \gg \text{replayRef } r$



- Expand reference: $\text{delay } 2 \gg \text{expandRef } (2*) r$



```
expandRef f r = do {  
  t0 ← now; a ← readRef r; d ← durRef r; t1 ← now  
  let d1 = f d - duration t1 t0 in case (d1 ≥ 0) of  
    True → delay d1  
    False → liftIO (print "Non causal shrink")  
  return a }
```


3. Monad streams

To achieve way more expressiveness

Higher-order streams

Lists extended with a type constructor for head/tail access.

```
newtype Stream f a
  = Stream { next :: f (Maybe (a, Stream f a)) }
```

with $next\ s :: f\ (Maybe\ (a, Stream\ f\ a))$ “evaluated” into:

- (1) either *Nothing* for the **terminated** stream ,
- (2) or *Just (a, sc)* for a **produced value a** and a **stream continuation sc**.

Timed IO streams

Timed IO streams

```
type STIO a = Stream TIO a
```

that essentially behaves like timed IO signals...

With good behavior

- (1) GC with full capacity to prevent memory leaks,
- (2) data flow programming with bounded memory usage,

solving some problems arising with FRP approach.

Example: timed standard IO with timed IO streams

Input as a timed IO stream:

```
streamInChar :: STIO Char  
streamInChar = Stream $ do  
  b ← liftIO $ hIsEOF stdin  
  if b then return Nothing  
  else do  
    a ← liftTimedIO getChar  
    return $ Just (a, streamInChar)
```

and streaming to output:

```
streamOutChar :: STIO Char → STIO ()  
streamOutChar (Stream m) = Stream $ do  
  c ← m  
  case c of  
    Nothing → return Nothing  
    Just (a, s) → do  
      liftIO $ putChar a  
      return $ Just ((), streamOutChar s)
```

Horizontal monoid structure

Putting streams **one after the other**:

```
instance Monad m  $\Rightarrow$  Monoid (Stream m a) where  
  mempty = Stream (return Nothing)  
  ( $\diamond$ ) (Stream m) s = Stream $ do  
    c  $\leftarrow$  m  
    case c of  
      Nothing  $\rightarrow$  next s  
      Just (a, sc)  $\rightarrow$  return $ Just (a, sc  $\diamond$  s)
```

with the second stream delayed for ever when first is infinite.

Vertical monoid structure

Merging streams by **local termination time**:

```
merge :: MonadRef m =>
  Stream m a → Stream m a → Stream m a
merge (Stream m1) (Stream m2) = Stream $ do
  r1 ← forkToRef m1
  r2 ← forkToRef m2
  c ← parReadRef r1 r2
  case c of
    Left Nothing → readRef r2
    Right Nothing → readRef r1
    Left (Just (a, mc1)) → return $
      Just (a, merge mc1 (Stream $ readRef r2))
    Right (Just (a, mc2)) → return $
      Just (a, merge (Stream $ readRef r1) mc2)
```

A resulting **associative** and, moreover, **commutative**, operator still with empty stream as neutral element.


Induced asynchronous monad

The induced (non standard) monad instance:

```
instance MonadRef m  $\Rightarrow$  Monad (Stream m) where  
  return a = (Stream  $\circ$  return  $\circ$  Just) (a, empty)  
  ( $\gg=$ ) (Stream m) f = Stream $ do  
    c  $\leftarrow$  m  
    case c of  
      Nothing  $\rightarrow$  return Nothing  
      Just (a, mc)  $\rightarrow$  next $ merge (f a) (mc  $\gg=$  f)
```

where the bind operation is defined by the merge of all parameterized monad streams from *when* there are produced !

The asynchronous bind of streams with monad references



4. Monad stream references

Or unbounded fifo chanel

Monad stream references

Monad streams can also be forked and referenced by:

```
type StreamRef m = Stream (Ref m)
```

with stream references reading given by:

```
readStreamRef :: MonadRef m =>  
  StreamRef m a → Stream m a  
readStreamRef (Stream v) = Stream $ do  
  c ← readRef v  
  case c of  
    Nothing → return Nothing  
    Just (a, rc) → return $ Just (a, readStreamRef rc)
```

Monad stream references

and streams forking into references given by:

```
forkStreamToRef :: MonadRef m =>
  Stream m a → m (StreamRef m a)
forkStreamToRef s = do
  r ← forkToRef (evalAndFork s)
  return $ Stream r
  where
evalAndFork (Stream m) = do
  c ← m
  case c of
    Nothing → return Nothing
    Just (a, sc) → do
      rc ← forkToRef (evalAndFork sc)
      return $ Just (a, Stream rc)
```

And more

Forking monad actions in traversable structures, and sorting them by termination time:

$$\begin{aligned} \text{forkAllToRefs} &:: \text{Traversable } t \Rightarrow \\ &t (m a) \rightarrow m (\text{StreamRef } m a) \end{aligned}$$

with an associated (linear time) on-the-fly folding

$$\begin{aligned} \text{foldStream} &:: \text{Monad } m \Rightarrow (b \rightarrow a \rightarrow m b) \rightarrow \\ &b \rightarrow \text{Stream } m a \rightarrow m b \\ \text{foldStream } f \ b \ (\text{Stream } m) &= m \gg\! = (\text{maybe } (\text{return } b) \\ &(\lambda(a, s) \rightarrow (f \ b \ a) \gg\! = \lambda b \rightarrow \text{foldStream } f \ b \ s)) \end{aligned}$$

--

$$\begin{aligned} \text{foldOnTime} &:: (\text{MonadStreamRef } m, \text{Traversable } t) \Rightarrow \\ &(b \rightarrow a \rightarrow m b) \rightarrow b \rightarrow t (m a) \rightarrow m b \\ \text{foldOnTime } f \ b \ c &= \\ &\text{forkAllToRefs } c \gg\! = \text{foldStream } f \ b \circ \text{readStreamRef} \end{aligned}$$

5. Conclusion

That's more than enough for now...

Conclusion

- ▶ We thus have defined:

Timed IO monad = *Timed monad* + *monad references*
+ *higher order streams* + *IO*

on top of **Haskell IO monad** + **Haskell concurrent library**.

- ▶ This extension provides **timed streams programming** while preserving GC capacity to keep memory bounded.

Future directions of research

- ▶ **Theoretical study** of monad references in CT ?
- ▶ **Static analysis tools** for detecting:
 - ▶ **time contraction** \Rightarrow non causal behavior,
 - ▶ **time expansion** \Rightarrow unbounded buffering,in timed IO stream functions ?
- ▶ **Safe interruption mechanism** for timed streams ?
- ▶ **Extension to hybrid system** : from timer to signal handler ?
- ▶ **Synchronous vs asynchronous** streaming ?
- ▶ and more **experiments**, e.g. audio processing and music. . .

Thanks for your attention

Any question ?